
@ Math Functions

@ABS(num)
Returns the Absolute Value of the number.
@ABS(-45.3) = 45.3
@ABS(@COS(PI)) = 1
@ABS(34) = 34
@ABS("Absolutely!") = 0

@ACOS(num)
Returns the Arccosine of num. ACOS(COS(num)) = num.
What is the angle with the cosine = 0.7071?
@ACOS(0.7071) = 0.7843 (radians): This is equivalent to an angle of 45 degrees.

@ACOS(0.7071)*180/@pi = 45: This gives the same answer as above, but in degrees.

@ASIN(num)
Returns the Arcsine of num. ASIN(SIN(num)) = num.
What is the angle whose sine = 0.3090?
@ASIN(0.3090) = 0.3142 (radians): This is equivalent to an angle of 18 degrees.

@ASIN(0.3090)*180/@pi = 18: This gives the same answer as above, but in degrees.

@ATAN(num)
Returns the Arctangent of num.
@ATAN(5.6713) = 1.3962 (radians): This is equivalent to an angle of 80 degrees.

@ATAN2(y-num,x-num)
Returns the Arctangent of x/y. This gives the angle of a line from the origin to x,y.
@ATAN2(3,1) = 0.3217 (radians): This is equivalent to an angle of 18.4 degrees.
This function differs from "ATAN2(x-num,y-num)" on page163. @ATAN2() requires the Y parameter first. ATAN2() 
requires the X parameter first.

@AVE(num|range[,...]) or @AVG(num|range[,...])
Divides the sum of the numbers by the count of numeric and string entries. Uses the same rules for counting as 
@COUNT().
@AVE(4,5,25,3) = 8.25

Given the following data:
A1: 184
A2: 592
A3: 97
@AVE(A1:A3) = 291
This function differs from "AVE(num|range[,...]) or AVERAGE(num|range[,...])" on page164. @AVE() and @AVG() 
treat string entries as having a zero value, but AVE() and AVERAGE() ignore string entries.



@COS(num)
Returns the Cosine of num.
@COS(@PI) = -1
@COS(@LN(3)) = 0.4548

@COUNT(num|range[,...])
Counts the number of numeric or string entries. Cells or items containing strings are counted. Cells or items 
containing blanks are not counted.

A B C
1 Duck 10
2 Soup 20
3 1.86 20
4 $30010
5 ======= 30
6 50
7 40% 70

@COUNT(A1:A7) = 6: There are only six items in column A because cell A6 is blank.

@COUNT(A1:A7,B1:B7) = 13: This counts the 6 items in column A and adds the 7 items in column B.

@COUNT("Good","Bad","Indifferent") = 3
This function differs from "COUNT(num|range[,...])" on page164. @COUNT() counts string entries. COUNT() 
ignores string entries and returns the count of cells that contain numeric values.

@EXP(num)
Returns e to the num power. EXP(LN(num)) = num.
@EXP(4) = 54.5982

+1/@EXP(12%*5) * 30000 = $16,464: This formula calculates the present value of $30,000 invested at 12% 
continually compounded interest for a term of 5 years.

@FRAC(num)
Returns the fractional component of num.
@FRAC(@NOW): This formula returns only the time component of the current date.

@FRAC(3.14159) = 0.14159

@INT(num)
Returns the integer portion of the number.
@INT(@RAND*10) : This formula returns a random number between 0 and 9.

A3 = 68293, @INT(@LOG(A3))+1 = 5: This formula tells how many digits are the number in cell A3.

@LN(num)



Returns the natural logarithm of num.
@LN(27) = 3.2958

B8 = 1024, @LN(B8)/@LN(2) = 10: This formula returns the base 2 log of cell B8.

@LOG(num)
Returns the logarithm, base 10, of num.
@LOG(45) = 1.6532

+10^(@LOG(292)) = 292: This formula takes the base 10 log of 292, then raises 10 to that power, resulting in the 
number it began with.

@MAX(num|range[,...])
Returns the largest number in the range or list. Strings are considered to be zero.
@MAX(4.5,3.2,2.5,2.5,6.2) = 6.2

@MAX(4,4,4) = 4

A B C
1 300 400
2 0 400
3 200 100
4 300 5000
5 700
6 600
7 300
8 500
@MAX(A1:A8,C1:C8) = 5000
This function differs from "MAX(num|range[,...])" on page166. @MAX() treats string values as zero, but MAX() 
ignores string values.

@MIN(num|range[,...])
Returns the smallest number in the range or list. String values are considered to be zero.
A1 = 300, A2 = 400, A3 = Hello!, @MIN(A1:A3) = 0:    This formula returns zero because the label "Hello!" 
evaluates to zero.

A1= 0.0002, A2 = 0, A3 = -339492, @MIN(A1:A3) = -339492
This function differs from "MIN(num|range[,...])" on page166. @MIN() treats string value as zero, but MIN() ignores
string values.

@MOD(num,div)
Returns the remainder of num divided by div
@MOD(12,5) = 2

@NOW-@MOD(@NOW,7)+5    :    This function returns the date-number for Monday of the current week..



@ROUND(num,prec)
Rounds num to prec decimal places. If prec is less than zero, rounds to the left of the decimal place.
@ROUND(350.2852,2) = 350.29
@ROUND(25492,-3) = 25000

@SIGN(num) or @SGN(num)
Returns 0 if num is zero, -1 if num is less than 0, and 1 if num is greater than 0.
B7 = @RAND, @SIGN(B7-0.5)    :    This formula returns the sign of a randomly generated number between 0.5 
and -0.5.

@SIGN(-405) = -1

@SIN(num)
Computes the Sine of the number. The number is expressed in radians.
@SIN(60*PI/180) = 0.8660    :    This formula gives the sine of 60 degrees.

A mountain road goes up at an incline of 25 degrees.    If the road is straight and is 20 miles long, what vertical 
distance will a car travel to climb it? +20 * @SIN(25*(@pi/180)) = 8.5    miles

@SQRT(num)
Returns the square root on the number.
@SQRT(34) = 5.8310
@SQRT(@LOG(200)) = 1.5169
@SQRT(@VAR(200,500,100)) = 170    :    This formula computes the standard deviation of the values given in the 
variance function.

@STD(num|range[,...])
Results in the standard deviation of the ranges or list. Strings are counted as 0, and blank cells are not counted.

A B
1 NAME SCORE
2 Anna 65.8
3 Bill 95.4
4 Donna 30.2
5 Mark54.9
6 Marie 35.1
7 Susan 75.9
8 John83.2
9 Rob 33.1
10 Ethan 81.8
@STD(B2:B10) = 23.1
This function differs from "STDDEV(num|range[,...]) or STDEV(num|range[,...])" on page167. @STD() counts 
strings as zero, but STDDEV() and STDEV() ignore string values.

@SUM(num|range[,...])
Adds all the numbers or sums the numbers in the range. You can specify as many numbers or ranges as you 
want.



@SUM(300,400,2300,100) = 3100

A1 = Blue, A2 = 300, A4 = 900, B1 = 1200, @SUM(A1:A4,B1) = 2400

@TAN(num)
Returns the Tangent of num.
@TAN(1.24) = 2.9119

@VAR(num|range[,...])
The result is the statistical variance of the numbers. Strings are counted as 0, and blank cells are not counted.

A B
1 NAME SCORE
2 Anna 65.8
3 Bill 95.4
4 Donna 30.2
5 Mark54.9
6 Marie 35.1
7 Susan 75.9
8 John83.2
9 Rob 33.1
10 Ethan 81.8
@VAR(B2:B10) = 531.6010
This function differs from "VAR(num|range[,...])" on page168. @VAR() counts string values as zero, but VAR() 
ignores string values.


